module IntSet = Set.Make (struct
  type t = int
  let compare = compare
end)

let rec upper_bound l r p =
  if r - l <= 1
  then l
  else let m = (l + r) / 2 in
       if p m 
       then upper_bound m r p
       else upper_bound l m p

let rec power ( * ) e m n =
  if n <= 0 then e
  else power ( * ) (if n land 1 = 0 then e else m * e) (m * m) (n lsr 1)

let code c = Char.code c - Char.code 'a'

exception Exists

let () = Scanf.scanf "%d\n%s" @@ fun n s ->
  Printf.printf "%d\n" @@
  upper_bound 0 (n / 2 + 1) @@ fun l ->
    Fun.flip List.for_all [1000000007; 1000000009] @@ fun modulo ->
      let ( +^ ) x y = (x + y) mod modulo in
      let ( -^ ) x y = x +^ (modulo - y) in
      let ( *^ ) x y = (x * y) mod modulo in
      let hash = Array.fold_left (fun h c -> 26 *^ h +^ code c) 0 in
      let radix = power ( *^ ) 1 26 l in
      let h = ref @@ hash @@ Array.init l @@ String.get s in
      let h' = ref @@ hash @@ Array.init l @@ fun i -> s.[i + l] in
      let set = ref @@ IntSet.singleton !h in
      try
        if IntSet.mem !h' !set then raise Exists;
        for i = 0 to n - 2 * l - 1 do
          h := 26 *^ !h +^ code s.[i + l] -^ radix *^ code s.[i];
          h' := 26 *^ !h' +^ code s.[i + 2 * l] -^ radix *^ code s.[i + l];
          set := IntSet.add !h !set;
          if IntSet.mem !h' !set then raise Exists;
        done; false
      with Exists -> true
